



DECUS

PROGRAM LIBRARY

DECUS NO.	8-222
TITLE	DISK MEMORY RETENTION TEST
AUTHOR	Edward A. Taft III
COMPANY	Saint Mark's School Southborough, Massachusetts
DATE	June 6, 1969
SOURCE LANGUAGE	PAL-D

DISK MEMORY RETENTION TEST

DECUS Program Library Write-up

DECUS No. 8-222

GENERAL

The DF32 Disk can sometimes drop bits in data written on it and left for a long period of time. There is currently no MAINDEC available that will allow a test pattern to be written and checked at a later time (after the computer has been off for a while, for example). The Disk Memory Retention Test is designed to allow this test to be made.

The test consists of three modes of operation: Write Data, Read Check, and Parity Check. The program is teletype operated, and error messages are printed. In Write mode, a user-selected pattern is written on the entire disk. In Read mode, the entire disk is read and tested for correct data and errors of any kind. In Parity mode, the entire disk is read and checked for errors only and not for data.

This program is not intended for diagnosis of any problem besides long-term memory retention.

REQUIREMENTS

This program runs on a PDP-8/I with teletype and DF32 Disk. It will not run on a PDP-8/S.

Core requirements: $\emptyset2000$ - $\emptyset777$.

LOADING

The binary tape is loaded into Field \emptyset with the Binary Loader. The starting address is $\emptyset2000$. There are no switch register options.

OPERATION

Load and start the program. It should respond by typing a right angle bracket.

There are three commands as follows:

Write

Press the W key; the program responds

> WRITE DATA:

and waits for the user to type an octal number, which is the data which is to be written.

Terminate the number by pressing Carriage-Return. The program will then write the specified pattern on the entire disk, responding with an angle when it is finished.

Read

Press the R key; the program responds

>READ CHECK:

and waits for the user to type the data with which the contents of the disk are to be compared. Again, terminate the number with a Carriage-Return. When it is finished checking, the program will type an angle.

Parity Check

Press the P key; the program will respond

> PARITY CHECK:

and check the entire disk for read errors.

Illegal typed characters are ignored and not echoed. If an error is made while typing a number, press RUBOUT and the program will restart.

DETAILS OF OPERATION

The program was designed to help track down dropped bits and parity errors which occur (especially with PDP-8/I systems) while the computer is turned off. The method of checking consists of writing a pattern on the disk, then running the program again several hours or days later to read check the data. The parity test mode was included to allow checking for parity errors and other read errors without wiping out the current contents of the disk by writing the single pattern.

The program may be restarted at location 200 as many times as desired.

This program tests only Unit 0; other units are not affected.

ERRORS

Write Errors

If an error occurs while the program is writing on the disk, the following message is typed:

WE 00000 T00 A0000

The first number typed is the full 15-bit disk address at which the error took place. This

address is split into a four-bit track number (after "T") and an eleven-bit address on the track (after "A"). All numbers are in octal.

When a write error takes place, the operation is aborted. If the error is not caused by (a) write lockout switches being on, or (b) no Unit 0 selected, the user should run the appropriate MAINDEC.

Read Errors

If an error occurs while the program is performing either a READ CHECK or a PARITY CHECK, the following message is typed:

```
E#0 00000 T00 A00000 D00000
```

The first number (the single digit after E#) is the error number. These are as follows:

E#0	There were no read errors, but the data is incorrect.
E#1	Parity error detected.
E#2	Nonexistent disk error.
E#4	Timing error.

These error numbers can be combined (like microinstructions); for example, E#5 would mean timing and parity errors. In general, E#2 and E#4 indicate problems that should be diagnosed by the appropriate MAINDEC.

The disk address, track number, and track address are printed as explained under Write Errors. The number following "D" is the data that was actually read from the disk at the location in question.

After a read error message has been printed, the program continues reading at the next sequential location. Read checking may be terminated at any time by striking CTRL/P on the teletype. The program will then type QUIT and restart.

RUNNING TIME

If there are no errors, the program takes approximately three seconds to write onto or read from the entire disk. If there are errors, the running time is the amount of time it takes to print the error messages.

EXAMPLE OF USE

```
> WRITE DATA: 7575  
WE 40000 T10 A00000
```

The upper 16K lockout switch was inadvertently left on.

```
>
```

```
> WRITE DATE: 7575
```

The write was completed successfully.

>
> READ CHECK: 7575 The disk was read and checked successfully.

>
> READ CHECK: 7575
E#1, 23145 T04 A3145 D7535 Two errors were detected this time, both consisting
E#0 56013 T13 A2012 D6535 of dropped bits.

>
> READ CHECK: 5757 The user specified the wrong number, and the
E#0 00000 T00 A0000 D7575 comparison test failed. The user interrupted
E#0 00001 T00 A0001 D7575 the typing by striking CTRL/P.
E#0 00002 T00 A0002 D7575
QUIT

>
> PARITY CHECK
E#1 23145 T04 A3145 D7535 A parity error was detected. The bad data at
> 56012 was not detected because the parity was
 good.

Note: It is not necessary to WRITE DATA before using PARITY CHECK. Anything whatsoever may be present on the disk; PARITY CHECK tests only for parity and read errors and not for correct data.

/DISK TEST PROGRAM EDWARD TAFT 5/7/69
 /ON COMMAND:
 /WRITES A SELECTED PATTERN ON THE ENTIRE DISK
 /READS AND CHECKS ENTIRE DISK, TO SEE THAT DATA
 /HAS BEEN RETAINED WITH NO ERRORS

*200

Ø2ØØ	6Ø32	START,	KCC	/INITIALIZE ALL
Ø2Ø1	6Ø42		TCF	
Ø2Ø2	66Ø1		DCMA	
Ø2Ø3	6611		DCEA	
Ø2Ø4	4253		JMS TXTOUT	/TYPE CR, LF, >
Ø2Ø5	Ø215		215	
Ø2Ø6	Ø212		212	
Ø2Ø7	4276		4276	
Ø21Ø	1377		TAD (-1Ø)	/INITIALIZE DISK WORD COUNT
Ø211	3776		DCA C2	
Ø212	3775		DCA C1	
Ø213	4774	GEK,	JMS KEYIN	/GET COMMAND CHARACTER FROM KEYBOARD
Ø214	1373		TAD (-215	/TEST FOR CR
Ø215	745Ø		SNA	
Ø216	52ØØ		JMP START	/YES, RESTART
Ø217	1372		TAD (215-327	/NO, TEST FOR WRITE
Ø22Ø	745Ø		SNA	
Ø221	5771		JMP WRITE	/YES, FILL DISK
Ø222	137Ø		TAD (327-322	/NO, TEST FOR READ
Ø223	745Ø		SNA	
Ø224	5767		JMP READ	/YES, READ-CHECK DATA
Ø225	1366		TAD (322-32Ø	/NO, PARITY TEST?
Ø226	764Ø		SZA CLA	
Ø227	5213		JMP GEK	/NO, IGNORE CHARACTER
Ø23Ø	4253		JMS TXTOUT	/YES, TYPE "PARITY CHECK"
Ø231	Ø32Ø		32Ø	
Ø232	Ø3Ø1		3Ø1	/THIS SECTION READS THE ENTIRE DISK BUT CHECK
Ø233	Ø322		322	/ONLY FOR READ ERRORS (PARITY, ETC.) AND
Ø234	Ø311		311	/DOES NOT CHECK FOR INCORRECT DATA
Ø235	Ø324		324	
Ø236	Ø331		331	
Ø237	Ø24Ø		24Ø	
Ø24Ø	Ø3Ø3		3Ø3	
Ø241	Ø31Ø		31Ø	
Ø242	Ø3Ø5		3Ø5	
Ø243	Ø3Ø3		3Ø3	
Ø244	Ø313		313	
Ø245	Ø215		215	
Ø246	Ø212		212	
Ø247	424Ø		424Ø	
Ø25Ø	1365		TAD (SKP CLA	/DISABLE DATA CHECKING
Ø251	3764		DCA CHK	
Ø252	5763		JMP DREAD	

		/PRINT CHARACTERS GIVEN IN ARG UNTIL NEG #	
Ø253	ØØØØ	TXTOUT, Ø	
Ø254	1653	TAD I TXTOUT	
Ø255	4762	JMS TTYOUT	/PRINT CHARACTER
Ø256	1653	TAD I TXTOUT	/TEST FOR NEGATIVE
Ø257	2253	ISZ TXTOUT	
Ø26Ø	77ØØ	SMA CLA	
Ø261	5254	JMP TXTOUT+1	/NO, CONTINUE
Ø262	5653	JMP I TXTOUT	/YES, DONE
		/READ IN AN OCTAL NUMBER	
Ø263	ØØØØ	NUMBER, Ø	
Ø264	3336	DCA OCNUM	/INITIALIZE
Ø265	4774	JMS KEYIN	/GET A CHARACTER
Ø266	1373	TAD (-215	/TEST FOR CR
Ø267	745Ø	SNA	
Ø27Ø	5311	JMP ENDNUM	/END OF #
Ø271	1361	TAD (215-267	/TEST FOR DIGIT
Ø272	754Ø	SMA SZA	
Ø273	5265	JMP NUMBER+2	/TOO LARGE, IGNORE
Ø274	136Ø	TAD (267-26Ø	
Ø275	751Ø	SPA	
Ø276	5265	JMP NUMBER+2	/TOO SMALL, IGNORE
Ø277	3337	DCA DIGIT	/OK, SAVE
Ø3ØØ	1337	TAD DIGIT	/ECHO
Ø3Ø1	1357	TAD (26Ø	
Ø3Ø2	4762	JMS TTYOUT	
Ø3Ø3	1336	TAD OCNUM	/ROTATE PREV NUMBER
Ø3Ø4	71Ø4	CLL RAL	/PREVENT OVERFLOW CARRYOVER
Ø3Ø5	71Ø4	CLL RAL	/ONLY LAST 4 DIGITS SIGNIFICANT
Ø3Ø6	71Ø4	CLL RAL	
Ø3Ø7	1337	TAD DIGIT	/ADD NEW DIGIT
Ø31Ø	5264	JMP NUMBER+1	
Ø311	4253	ENDNUM, JMS TXTOUT	/PRINT CR, LF, SPACE
Ø312	Ø215	215	
Ø313	Ø212	212	
Ø314	424Ø	424Ø	
Ø315	1336	TAD OCNUM	/EXIT WITH NUMBER IN AC
Ø316	5663	JMP I NUMBER	

/PRINT AN N-DIGIT OCTAL NUMBER

/NUMBER IN AC, -N IN LOC "N"

Ø317	ØØØØ	OCOUT, Ø	
Ø32Ø	3336	DCA OCNUM	
Ø321	1336	TAD OCNUM	/GET A DIGIT
Ø322	Ø356	AND (7ØØØ	
Ø323	71Ø6	CLL RTL	
Ø324	7ØØ6	RTL	
Ø325	1357	TAD (26Ø	
Ø326	4762	JMS TTYOUT	/PRINT IT
Ø327	1336	TAD OCNUM	
Ø33Ø	71Ø6	CLL RTL	/ROTATE NUMBER
Ø331	7ØØ4	RAL	
Ø332	3336	DCA OCNUM	
Ø333	234Ø	ISZ N	/CHECK COUNT
Ø334	5321	JMP OCOUT+2	
Ø335	5717	JMP I OCOUT	
Ø336	ØØØØ	OCNUM, Ø	
Ø337	ØØØØ	DIGIT, Ø	
Ø34Ø	ØØØØ	N, Ø	
Ø356	7ØØØ		
Ø357	Ø26Ø		
Ø36Ø	ØØØ7		
Ø361	7726		
Ø362	Ø454		
Ø363	Ø622		
Ø364	Ø642		
Ø365	761Ø		
Ø366	ØØØ2		
Ø367	Ø6ØØ		
Ø37Ø	ØØØ5		
Ø371	Ø4ØØ		
Ø372	7666		
Ø373	7563		
Ø374	Ø462		
Ø375	Ø737		
Ø376	Ø74Ø		
Ø377	777Ø		

PAGE

		/WRITE ON DISK	
Ø4ØØ	4777	WRITE, JMS TXTOUT	/PRINT "WRITE DATA:"
Ø4Ø1	Ø327	327	
Ø4Ø2	Ø322	322	
Ø4Ø3	Ø311	311	
Ø4Ø4	Ø324	324	
Ø4Ø5	Ø3Ø5	3Ø5	
Ø4Ø6	Ø24Ø	24Ø	
Ø4Ø7	Ø3Ø4	3Ø4	
Ø41Ø	Ø3Ø1	3Ø1	
Ø411	Ø324	324	
Ø412	Ø3Ø1	3Ø1	
Ø413	Ø272	272	
Ø414	424Ø	424Ø	
Ø415	4776	JMS NUMBER	/GET OCTAL NUMBER
Ø416	3336	DCA WORD	
Ø417	3775	DCA WC	/SET WC AND CA REGISTERS
Ø42Ø	1374	TAD (WORD-1	
Ø421	3773	DCA CA	
Ø422	66Ø5	DMAW	/START WRITING
Ø423	6621	WLOOP, DFSE	/CHECK FOR WRITE ERROR
Ø424	5244	JMP WERROR	
Ø425	1775	TAD WC	/WAIT FOR ONE TRANSFER
Ø426	765Ø	SNA CLA	
Ø427	5223	JMP WLOOP	
Ø43Ø	3775	DCA WC	/DONE, RESET REGISTERS
Ø431	1374	TAD (WORD-1	
Ø432	3773	DCA CA	
Ø433	6621	DFSE	/CHECK AGAIN FOR ERRORS
Ø434	5244	JMP WERROR	
Ø435	2772	ISZ C1	/CHECK WORD COUNT
Ø436	5223	JMP WLOOP	
Ø437	2771	ISZ C2	
Ø44Ø	5223	JMP WLOOP	
Ø441	1374	TAD (WORD-1	
Ø442	3773	DCA CA	
Ø443	577Ø	JMP DEND	/WRITE FINISHED, CLEAR ALL REGISTERS
		/WRITE ERROR FOUND	
Ø444	724Ø	WERROR, CLA CMA	/PREMATURELY END WRITE
Ø445	3775	DCA WC	/AFTER NEXT WORD
Ø446	4777	JMS TXTOUT	/PRINT "WE"
Ø447	Ø327	327	
Ø45Ø	Ø3Ø5	3Ø5	
Ø451	424Ø	424Ø	
Ø452	4273	JMS ERPRINT	/PRINT ERROR ADDRESS
Ø453	5767	JMP START	

		/PRINT A CHARACTER IN AC	
Ø454	ØØØØ	TTYOUT, Ø	
Ø455	6Ø46	TLS	
Ø456	6Ø41	TSF	
Ø457	5256	JMP .-1	
Ø46Ø	72ØØ	CLA	
Ø461	5654	JMP I TTYOUT	
		/READ A CHARACTER FROM THE KEYBOARD	
Ø462	ØØØØ	KEYIN, Ø	
Ø463	6Ø31	KSF	
Ø464	5263	JMP .-1	
Ø465	6Ø36	KRB	
Ø466	1366	TAD (-377	/TEST FOR RUBOUT
Ø467	745Ø	SNA	
Ø47Ø	5767	JMP START	/RUBOUT TYPED, RESTART
Ø471	1365	TAD (377	
Ø472	5662	JMP I KEYIN	

/ROUTINE TO PRINT ERROR ADDRESS

/ XXXXX TXX AXXXX

/FULL ADDRESS TRACK # ADDRESS ON TRACK

Ø473	ØØØØ	ERPRNT, Ø	
Ø474	734Ø	CLL CLA CMA	/PRINT FULL ADDRESS FIRST
Ø475	3764	DCA N	
Ø476	1772	TAD C1	/GET LAST TRACK BIT (DMAØ)
Ø477	77ØØ	SMA CLA	/COMPLEMENT INTO LINK
Ø5ØØ	7Ø2Ø	CML	
Ø5Ø1	1771	TAD C2	/GET EXTENDED BITS-1Ø
Ø5Ø2	1363	TAD (1Ø)	/THIS WILL ALSO COMPLEMENT LINK
Ø5Ø3	Ø362	AND (7	/TRACK # NOW IN AC9-LINK
Ø5Ø4	7Ø12	RTR	
Ø5Ø5	7Ø12	RTR	
Ø5Ø6	3337	DCA TRACK	/NOW IN ACØ-AC3
Ø5Ø7	1337	TAD TRACK	
Ø51Ø	4761	JMS OCOUT	/PRINT FIRST ADDRESS DIGIT
Ø511	136Ø	TAD (-4	
Ø512	3764	DCA N	
Ø513	1772	TAD C1	/PRINT LAST 4 ADDRESS DIGITS
Ø514	4761	JMS OCOUT	
Ø515	4777	JMS TXTOUT	/PRINT "T" FOR TRACK
Ø516	Ø24Ø	24Ø	
Ø517	4324	4324	
Ø52Ø	7344	CLL CLA CMA RAL	
Ø521	3764	DCA N	
Ø522	1337	TAD TRACK	
Ø523	7112	CLL RTR	
Ø524	4761	JMS OCOUT	/PRINT 2-DIGIT TRACK #
Ø525	4777	JMS TXTOUT	/PRINT "A" FOR ADDRESS ON TRACK
Ø526	Ø24Ø	24Ø	
Ø527	43Ø1	43Ø1	
Ø53Ø	136Ø	TAD (-4	
Ø531	3764	DCA N	
Ø532	715Ø	CLL CMA RAR	/=3777
Ø533	Ø772	AND C1	/GET DMA1-11
Ø534	4761	JMS OCOUT	/PRINT 4-DIGIT ADDRESS
Ø535	5673	JMP I ERPRNT	

WORD, Ø

TRACK, Ø

WC=775Ø

CA=7751

/DATA BREAK LOCATIONS

Ø56Ø	7774
Ø561	Ø317
Ø562	ØØØ7
Ø563	ØØ1Ø
Ø564	Ø34Ø
Ø565	Ø377
Ø566	74Ø1
Ø567	Ø2ØØ

0570	0650
0571	0740
0572	0737
0573	7751
0574	0535
0575	7750
0576	0263
0577	0253

PAGE

		/READ-CHECK THE DATA	
Ø6ØØ	4777	READ, JMS TXTOUT	/TYPE "READ CHECK:"
Ø6Ø1	Ø322	322	
Ø6Ø2	Ø3Ø5	3Ø5	
Ø6Ø3	Ø3Ø1	3Ø1	
Ø6Ø4	Ø3Ø4	3Ø4	
Ø6Ø5	Ø24Ø	24Ø	
Ø6Ø6	Ø3Ø3	3Ø3	
Ø6Ø7	Ø31Ø	31Ø	
Ø61Ø	Ø3Ø5	3Ø5	
Ø611	Ø3Ø3	3Ø3	
Ø612	Ø313	313	
Ø613	Ø272	272	
Ø614	424Ø	424Ø	
Ø615	4776	JMS NUMBER	/GET OCTAL NUMBER
Ø616	7Ø41	CIA	/SHOULD BE WHAT WAS PREVIOUSLY
Ø617	3336	DCA CHECK	/WRITTEN ON DISK
Ø62Ø	1375	TAD (SZA CLA	/ENABLE DATA CHECKING
Ø621	3242	DCA CHK	
Ø622	66Ø3	DREAD, DMAR	/ENABLE DISK READ
Ø623	1374	RNEXT, TAD (DATA-1	/SET UP WC AND CA
Ø624	3773	DCA CA	
Ø625	3772	DCA WC	
Ø626	3342	DCA DATA	/ZERO OUT DATA CELL
Ø627	3341	DCA TIME	/SET UP TIMER
Ø630	1772	RLOOP, TAD WC	/WAIT FOR TRANSFER
Ø631	764Ø	SZA CLA	
Ø632	5236	JMP ROUT	/TRANSFER DONE
Ø633	2341	ISZ TIME	/NOT DONE, CHECK TIMING
Ø634	523Ø	JMP RLOOP	
Ø635	5256	JMP RERROR	/43 MSEC UP, SOMETHING'S WRONG
Ø636	6621	ROUT, DF SE	/CHECK FOR ERRORS
Ø637	5256	JMP RERROR	
Ø64Ø	1342	TAD DATA	/TEST DATA JUST READ
Ø641	1336	TAD CHECK	
Ø642	764Ø	CHK, SZA CLA	/(SKP CLA) FOR NO DATA CHECKING
Ø643	5256	JMP RERROR	/DID NOT MATCH
Ø644	2337	ISZ C1	/OK, CHECK WORD COUNT
Ø645	5223	JMP RNEXT	
Ø646	234Ø	ISZ C2	
Ø647	5223	JMP RNEXT	
Ø65Ø	724Ø	DEND, CLA CMA	/TRANSFER COMPLETE, STOP DISK
Ø651	3772	DCA WC	
Ø652	3341	DCA TIME	
Ø653	2341	ISZ TIME	/WAIT FOR EVERYTHING TO STOP
Ø654	5253	JMP .-1	
Ø655	5771	JMP START	

		/DATA DOES NOT CHECK, OR DISK ERROR INDICATED	
Ø656	724Ø	RERROR, CLA CMA	/STOP TRANSFERS UNTIL ERROR
Ø657	3772	DCA WC	/MESSAGE PRINTED
Ø66Ø	6616	DEAC	/GET ERROR BITS
Ø661	Ø37Ø	AND (7	
Ø662	1367	TAD (26Ø	/CONSTRUCT ERROR #
Ø663	3267	DCA .+4	
Ø664	4777	JMS TXTOUT	/PRINT "E#N" WHERE N IS ERROR
Ø665	Ø3Ø5	3Ø5	
Ø666	Ø243	243	
Ø667	ØØØØ	Ø	/ERROR NUMBER GOES HERE
Ø67Ø	424Ø	424Ø	
Ø671	4766	JMS ERPRINT	/PRINT ERROR ADDRESS
Ø672	4777	JMS TXTOUT	/PRINT "D" FOR DATA
Ø673	Ø24Ø	24Ø	
Ø674	43Ø4	43Ø4	
Ø675	1365	TAD (-4	
Ø676	3764	DCA N	
Ø677	1342	TAD DATA	
Ø7ØØ	4763	JMS OCOUT	/PRINT DATA
Ø7Ø1	4777	JMS TXTOUT	/CR, LF, SPACE
Ø7Ø2	Ø215	215	
Ø7Ø3	Ø212	212	
Ø7Ø4	424Ø	424Ø	
Ø7Ø5	6Ø37	KSF KRB	/CHECK FOR KEY INTERRUPT
Ø7Ø6	72ØØ	CLA	/CTRL/P
Ø7Ø7	1362	TAD (-22Ø	
Ø71Ø	765Ø	SNA CLA	
Ø711	533Ø	JMP INT	/YES, INTERRUPTED
Ø712	2337	ISZ C1	/INCREMENT WORD COUNT
Ø713	741Ø	SKP	
Ø714	234Ø	ISZ C2	
Ø715	741Ø	SKP	
Ø716	5771	JMP START	/THAT WAS LAST WORD
Ø717	134Ø	TAD C2	/RELOAD EXTENDED REGISTER
Ø72Ø	1361	TAD (1Ø	
Ø721	71Ø6	CLL RTL	
Ø722	7ØØ6	RTL	
Ø723	7ØØ6	RTL	
Ø724	6615	DEAL	
Ø725	72ØØ	CLA	
Ø726	1337	TAD C1	/RELOAD ADDRESS REGISTER
Ø727	5222	JMP DREAD	/AND CONTINUE
Ø73Ø	4777	JMS TXTOUT	/INTERRUPTED VIA CTRL/P
Ø731	Ø321	321	
Ø732	Ø325	325	
Ø733	Ø311	311	
Ø734	4324	4324	
Ø735	5771	JMP START	
Ø736	ØØØØ	CHECK, Ø	

0737	0000	C1,	0
0740	0000	C2,	0
0741	0000	TIME,	0
0742	0000	DATA,	0

/LIST OF ERROR MESSAGES
 /THEY MAY ALSO BE COMBINED LIKE MICROINSTRUCTIONS
 /
 /E#0 NO DISK ERROR, BUT DATA IS INCORRECT
 /E#1 PARITY ERROR
 /E#2 NONEXISTENT DISK
 /E#4 TIMING ERROR
 /
 /FOR EXAMPLE: E#5 WOULD MEAN TIMING AND PARITY ERRORS
 /
 /THIS PROGRAM IS DESIGNED TO BE USED ONLY WHEN
 /RETENTION PROBLEMS OCCUR AND THE MAINDECS SHOW UP
 /NO ERRORS. E#2 AND E#4, IF THEY OCCUR, ARE PROBLEMS
 /WHICH MAY BE DIAGNOSED BY "DISK DATA" MAINDEC. DITTO
 /WITH "WE" (UNLESS YOU LEFT ON THE WRITE LOCK!). THE
 /ONLY USE FOR THIS PROGRAM IS TO TRACK DOWN ERRORS
 /THAT OCCUR OVER LONG PERIODS OF TIME OR WHILE THE
 /COMPUTER IS OFF.

/
REV. 6/5/69

0761	0010
0762	7560
0763	0317
0764	0340
0765	7774
0766	0473
0767	0260
0770	0007
0771	0200
0772	7750
0773	7751
0774	0741
0775	7640
0776	0263
0777	0253

CA	7751
CHECK	Ø736
CHK	Ø642
C1	Ø737
C2	Ø74Ø
DATA	Ø742
DEND	Ø65Ø
DIGIT	Ø337
DREAD	Ø622
ENDNUM	Ø311
ERPRNT	Ø473
GEK	Ø213
INT	Ø73Ø
KEYIN	Ø462
N	Ø34Ø
NUMBER	Ø263
OCNUM	Ø336
OCOUT	Ø317
READ	Ø6ØØ
RERROR	Ø656
RLOOP	Ø630
RNEXT	Ø623
ROUT	Ø636
START	Ø2ØØ
TIME	Ø741
TRACK	Ø537
TTYOUT	Ø454
TXTOUT	Ø253
WC	775Ø
WERROR	Ø444
WLOOP	Ø423
WORD	Ø536
WRITE	Ø4ØØ